NAG Fortran Library Routine Document

F08UCF (DSBGVD)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08UCF (DSBGVD) computes all the eigenvalues, and optionally, the eigenvectors of a real generalized symmetric-definite banded eigenproblem, of the form

$$Az = \lambda Bz,$$

where A and B are symmetric and banded, and B is also positive-definite. If eigenvectors are desired, it uses a divide-and-conquer algorithm.

2 Specification

```
 SUBROUTINE F08UCF (JOBZ, UPLO, N, KA, KB, AB, LDAB, BB, LDBB, W, Z, LDZ,
1                   WORK, LWORK, IWORK, LIWORK, INFO)
 INTEGER            N, KA, KB, LDAB, LDBB, LDZ, LWORK, IWORK(*), LIWORK,
1                   INFO
 double precision   AB(LDAB,*), BB(LDBB,*), W(*), Z(LDZ,*), WORK(*)
 CHARACTER*1        JOBZ, UPLO
```

The routine may be called by its LAPACK name ***dsbgvd***.

3 Description

The generalized symmetric-definite band problem

$$Az = \lambda Bz$$

is first reduced to a standard band symmetric problem

$$Cx = \lambda x,$$

where C is a symmetric band matrix, using Wilkinson's modification to Crawford's algorithm (see Crawford (1973) and Wilkinson (1977)). The symmetric eigenvalue problem is then solved for the eigenvalues and the eigenvectors, if required, which are then backtransformed to the eigenvectors of the original problem.

The eigenvectors are normalized so that the matrix of eigenvectors, Z, satisfies

$$Z^T AZ = \Lambda \quad \text{and} \quad Z^T BZ = I,$$

where Λ is the diagonal matrix whose diagonal elements are the eigenvalues.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

Crawford C R (1973) Reduction of a band-symmetric generalized eigenvalue problem *Comm. ACM* **16** 41–44

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Wilkinson J H (1977) Some recent advances in numerical linear algebra *The State of the Art in Numerical Analysis* (ed D A H Jacobs) Academic Press

5 Parameters

1: JOBZ – CHARACTER*1 *Input*

On entry: if JOBZ = 'N', compute eigenvalues only.

If JOBZ = 'V', compute eigenvalues and eigenvectors.

Constraint: JOBZ = 'N' or 'V'.

2: UPLO – CHARACTER*1 *Input*

On entry: if UPLO = 'U', the upper triangles of A and B are stored.

If UPLO = 'L', the lower triangles of A and B are stored.

3: N – INTEGER *Input*

On entry: n, the order of the matrices A and B.

Constraint: $\mathrm{N} \geq 0$.

4: KA – INTEGER *Input*

On entry: ka, the number of super-diagonals of the matrix A if UPLO = 'U', or the number of sub-diagonals if UPLO = 'L'.

Constraint: $\mathrm{KA} \geq 0$.

5: KB – INTEGER *Input*

On entry: kb, the number of super-diagonals of the matrix B if UPLO = 'U', or the number of sub-diagonals if UPLO = 'L'.

Constraint: $\mathrm{KB} \geq 0$.

6: AB(LDAB,$*$) – ***double precision*** array *Input/Output*

Note: the second dimension of the array AB must be at least $\max(1, \mathrm{N})$.

On entry: the upper or lower triangle of the symmetric band matrix A, stored in the first $ka + 1$ rows of the array. The jth column of A is stored in the jth column of the array AB as follows:

if UPLO = 'U', $\mathrm{AB}(ka + 1 + i - j, j) = a_{ij}$ for $\max(1, j - ka) \leq i \leq j$;
if UPLO = 'L', $\mathrm{AB}(1 + i - j, j) = a_{ij}$ for $j \leq i \leq \min(n, j + ka)$.

On exit: the contents of AB are destroyed.

7: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F08UCF (DSBGVD) is called.

Constraint: $\mathrm{LDAB} \geq \mathrm{KA} + 1$.

8: BB(LDBB,$*$) – ***double precision*** array *Input/Output*

Note: the second dimension of the array BB must be at least $\max(1, \mathrm{N})$.

On entry: the upper or lower triangle of the symmetric band matrix B, stored in the first $kb + 1$ rows of the array. The jth column of B is stored in the jth column of the array BB as follows:

if UPLO = 'U', $\mathrm{BB}(kb + 1 + i - j, j) = b_{ij}$ for $\max(1, j - kb) \leq i \leq j$;
if UPLO = 'L', $\mathrm{BB}(1 + i - j, j) = b_{ij}$ for $j \leq i \leq \min(n, j + kb)$.

On exit: the factor S from the split Cholesky factorization $B = S^T S$, as returned by F08UFF (DPBSTF).

9: LDBB – INTEGER *Input*

On entry: the first dimension of the array BB as declared in the (sub)program from which F08UCF (DSBGVD) is called.

Constraint: $\mathrm{LDBB} \geq \mathrm{KB} + 1$.

10: $\mathrm{W}(*)$ – ***double precision*** array *Output*

Note: the dimension of the array W must be at least $\max(1, \mathrm{N})$.

On exit: if $\mathrm{INFO} = 0$, the eigenvalues in ascending order.

11: $\mathrm{Z}(\mathrm{LDZ},*)$ – ***double precision*** array *Output*

Note: the second dimension of the array Z must be at least $\max(1, \mathrm{N})$.

On exit: if JOBZ = 'V', then if $\mathrm{INFO} = 0$, Z contains the matrix Z of eigenvectors, with the ith column of Z holding the eigenvector associated with $\mathrm{W}(i)$. The eigenvectors are normalized so that $Z^T BZ = I$.

If JOBZ = 'N', Z is not referenced.

12: LDZ – INTEGER *Input*

On entry: the first dimension of the array Z as declared in the (sub)program from which F08UCF (DSBGVD) is called.

Constraints:

if JOBZ = 'V', $\mathrm{LDZ} \geq \max(1, \mathrm{N})$;
$\mathrm{LDZ} \geq 1$ otherwise.

13: $\mathrm{WORK}(*)$ – ***double precision*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, \mathrm{LWORK})$.

On exit: if $\mathrm{INFO} = 0$, $\mathrm{WORK}(1)$ returns the optimal LWORK.

14: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08UCF (DSBGVD) is called.

If $\mathrm{LWORK} = -1$, a workspace query is assumed; the routine only calculates the optimal sizes of the WORK and IWORK arrays, returns these values as the first entries of the WORK and IWORK arrays, and no error message related to LWORK or LIWORK is issued.

Constraints:

if $\mathrm{N} \leq 1$, $\mathrm{LWORK} \geq 1$;
if JOBZ = 'N' and $\mathrm{N} > 1$, $\mathrm{LWORK} \geq \max(1, 3 \times \mathrm{N})$;
if JOBZ = 'V' and $\mathrm{N} > 1$, $\mathrm{LWORK} \geq 1 + 5 \times \mathrm{N} + 2 \times \mathrm{N}^2$.

15: $\mathrm{IWORK}(*)$ – INTEGER array *Workspace*

Note: the dimension of the array IWORK must be at least $\max(1, \mathrm{LIWORK})$.

On exit: if $\mathrm{INFO} > 0$, $\mathrm{IWORK}(1)$ returns the optimal LIWORK.

16: LIWORK – INTEGER *Input*

On entry: the dimension of the array IWORK as declared in the (sub)program from which F08UCF (DSBGVD) is called.

If $\mathrm{LIWORK} = -1$, a workspace query is assumed; the routine only calculates the optimal sizes of the WORK and IWORK arrays, returns these values as the first entries of the WORK and IWORK arrays, and no error message related to LWORK or LIWORK is issued.

Constraints:

if JOBZ $=$ 'N' or $\mathrm{N} \leq 1$, LIWORK ≥ 1;
if JOBZ $=$ 'V' and $\mathrm{N} > 1$, LIWORK $\geq 3 + 5 \times \mathrm{N}$.

17: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith argument had an illegal value.

INFO > 0

If INFO $= i$ and $i \leq \mathrm{N}$, the algorithm failed to converge: i off-diagonal elements of an intermediate tridiagonal form did not converge to zero.

If INFO $= i$ and $i > \mathrm{N}$, if INFO $= \mathrm{N} + i$, for $1 \leq i \leq \mathrm{N}$, then F08UFF (DPBSTF) returned 'INFO $= i$: B is not positive-definite'. The factorization of B could not be completed and no eigenvalues or eigenvectors were computed.

7 Accuracy

If B is ill-conditioned with respect to inversion, then the error bounds for the computed eigenvalues and vectors may be large, although when the diagonal elements of B differ widely in magnitude the eigenvalues and eigenvectors may be less sensitive than the condition of B would suggest. See Section 4.10 of Anderson *et al.* (1999) for details of the error bounds.

8 Further Comments

The total number of floating point operations is proportional to n^3 if JOBZ $=$ 'V' and, assuming that $n \gg k_a$, is approximately proportional to $n^2 k_a$ otherwise.

The complex analogue of this routine is F08UQF (ZHBGVD).

9 Example

To find all the eigenvalues of the generalized band symmetric eigenproblem $Az = \lambda Bz$, where

$$A = \begin{pmatrix} 0.24 & 0.39 & 0.42 & 0 \\ 0.39 & -0.11 & 0.79 & 0.63 \\ 0.42 & 0.79 & -0.25 & 0.48 \\ 0 & 0.63 & 0.48 & -0.03 \end{pmatrix}$$

and

$$B = \begin{pmatrix} 2.07 & 0.95 & 0 & 0 \\ 0.95 & 1.69 & -0.29 & 0 \\ 0 & -0.29 & 0.65 & -0.33 \\ 0 & 0 & -0.33 & 1.17 \end{pmatrix}.$$

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F08UCF Example Program Text
*     Mark 21.  NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, KAMAX, KBMAX
      PARAMETER        (NMAX=20,KAMAX=5,KBMAX=5)
      INTEGER          LDAB, LDBB, LIWORK, LWORK
      PARAMETER        (LDAB=KAMAX+1,LDBB=KBMAX+1,LIWORK=1,LWORK=3*NMAX)
      CHARACTER        UPLO
      PARAMETER        (UPLO='U')
*     .. Local Scalars ..
      INTEGER          I, INFO, J, KA, KB, N
*     .. Local Arrays ..
      DOUBLE PRECISION AB(LDAB,NMAX), BB(LDBB,NMAX), DUMMY(1,1),
     +                 W(NMAX), WORK(LWORK)
      INTEGER          IWORK(LIWORK)
*     .. External Subroutines ..
      EXTERNAL         DSBGVD
*     .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08UCF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, KA, KB
      IF (N.LE.NMAX .AND. KA.LE.KAMAX .AND. KB.LE.KBMAX) THEN
*
*        Read the upper or lower triangular parts of the matrices A and
*        B from data file
*
         IF (UPLO.EQ.'U') THEN
            READ (NIN,*) ((AB(KA+1+I-J,J),J=I,MIN(N,I+KA)),I=1,N)
            READ (NIN,*) ((BB(KB+1+I-J,J),J=I,MIN(N,I+KB)),I=1,N)
         ELSE IF (UPLO.EQ.'L') THEN
            READ (NIN,*) ((AB(1+I-J,J),J=MAX(1,I-KA),I),I=1,N)
            READ (NIN,*) ((BB(1+I-J,J),J=MAX(1,I-KB),I),I=1,N)
         END IF
*
*        Solve the generalized symmetric band eigenvalue problem
*        A*x = lambda*B*x
*
         CALL DSBGVD('No vectors',UPLO,N,KA,KB,AB,LDAB,BB,LDBB,W,DUMMY,
     +               1,WORK,LWORK,IWORK,LIWORK,INFO)
*
         IF (INFO.EQ.0) THEN
*
*           Print solution
*
            WRITE (NOUT,*) 'Eigenvalues'
            WRITE (NOUT,99999) (W(J),J=1,N)
         ELSE IF (INFO.GT.N .AND. INFO.LE.2*N) THEN
            I = INFO - N
            WRITE (NOUT,99998) 'The leading minor of order ', I,
     +        ' of B is not positive definite'
         ELSE
            WRITE (NOUT,99997) 'Failure in DSBGVD. INFO =', INFO
         END IF
      ELSE
         WRITE (NOUT,*) 'NMAX too small'
      END IF
      STOP
*
99999 FORMAT (3X,(6F11.4))
```

```
99998 FORMAT (1X,A,I4,A)
99997 FORMAT (1X,A,I4)
      END
```

9.2 Program Data

```
F08UCF Example Program Data

  4      2      1             :Values of N, KA and KB

  0.24   0.39   0.42
        -0.11   0.79   0.63
               -0.25   0.48
                      -0.03   :End of matrix A
  2.07   0.95
         1.69  -0.29
                0.65  -0.33
                       1.17   :End of matrix B
```

9.3 Program Results

```
 F08UCF Example Program Results

 Eigenvalues
       -0.8305    -0.6401     0.0992     1.8525
```